



Limites multidirectionnelles

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2001. L'objectif principal de ce document Maple est d'illustrer formellement la limite d'une fonction réelle de deux variables réelles. Dans cette feuille Maple, on réalisera que parfois il est nécessaire de travailler un peu plus pour obtenir un rendu plus conforme du tracé d'une surface dans l'espace.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,display,spacecurve,setoptions3d):
> setoptions3d(labels=[x,y,z], size=[300,300],
               tickmarks=[6,6,12],linestyle=0,
               axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8]);
> Opts3D:= grid=[60,60],lightmodel=none,shading=zhue:
> macro(domaine="Niagara 16"):
```


Cas où la limite existe

Soit la fonction f définie par $f(x,y) = 10 + xy^2$.

On a que $\text{dom}_f = \mathbb{R}^2$.

Traçons cette fonction sur un pavé circulaire centré à l'origine de rayon 5, soit pour $x \in [-5, 5]$ et pour $y \in [-\sqrt{25-x^2}, \sqrt{25-x^2}]$.

```
> f:=(x,y)->10 +x*y^2;
                                     f := (x,y) ↦ 10 + x·y2
> Surface:=plot3d([x,y,f(x,y)],x=-5..5,y=-sqrt(25-x^2)..sqrt(25-x^2),
                 Opts3D):
Domaine:=plot3d([x,y,-40],x=-5..5,y=-sqrt(25-x^2)..sqrt(25-x^2),
               style=wireframe,
               color=domaine,grid=[25,25],transparency=0.8):
> display(Surface,Domaine,axes=frame,orientation=[45,80]);
```




Sélectionnez le graphique et en tenant le bouton gauche de la souris pressé, modifiez l'orientation du graphique. La perception de la troisième dimension sera à son maximum.

Illustrons graphiquement l'existence de la limite de cette fonction lorsque $(x, y) \rightarrow (1, 2)$.

Traçons à nouveau cette fonction mais sur un pavé rectangulaire pour illustrer notre propos. Soit donc pour $x \in [-2, 3]$ et pour $y \in [-2, 3]$.

```
> Surface:=plot3d([x,y,f(x,y)],x=-2..3,y=-2..3,Opts3D):
  Domaine:=plot3d([x,y,0],x=-2..3,y=-2..3,style=wireframe,
    color=domaine,grid=[25,25],transparency=0.8):
> display(Surface,Domaine,axes=frame,orientation=[30,70]);
```



Puisque $dom_f = \mathbb{R}^2$, quelque soit le voisinage de $(1, 2)$, la fonction f est toujours définie.

Superposons au graphique précédent, un voisinage circulaire de $(1, 2)$ et réalisons le tracé de l'image de ce voisinage:


- ce voisinage sera visualisé par un cercle centré au point $(1, 2, 0)$ dans le plan XY
- l'image de ce voisinage sera visualisé par la portion intérieure de la surface limitée par l'image du périmètre du voisinage de $(1, 2, 0)$.

```
> Voisinage:=spacecurve([1+cos(t)/4,2+sin(t)/4,0],t=0..2*Pi,color=
  "Niagara 1",thickness=2):
  Image:=spacecurve([1+cos(t)/4,2+sin(t)/4,f(1+cos(t)/4,2+sin(t)/4)],t=
  0..2*Pi,
    color="Niagara 2",thickness=2):
  Horizontale_domaine:=spacecurve([1,t,0],t=0..2,color=black,thickness=
```

```

0,linestyle=2):
Verticale_domaine:=spacecurve([t,2,0],t=0..1,color=black,thickness=0,
linestyle=2):
> display([Surface,Domaine,Voisinage,Image,Horizontale_domaine,
Verticale_domaine],orientation=[25,72]);

```




Illustrons maintenant la projection de ces images sur le plan XZ et bornons, dans le plan XZ, la cote de ces images entre les traces des plans d'équations $z = 14 - \epsilon$ et $z = 14 + \epsilon$ en prenant $\epsilon = 2$.

```


> Proj_xz:=spacecurve([1+cos(t)/4,0,f(1+cos(t)/4,2+sin(t)/4)],t=0..2*
Pi,
color="Niagara 2",thickness=2):
Verticale:=spacecurve([1,2,t],t=0..f(1,2),linestyle=2,thickness=1,
color=orange):
Horizontale:=spacecurve([1,t,f(1,2)],t=0..2,linestyle=2,thickness=1,
color=orange):
L:=spacecurve([t,0,14],t=0..3,thickness=2,color=orange,linestyle=2):
epsilon:=2:
L_moins_delta:=spacecurve([t,0,14-epsilon],t=0..3,thickness=2,color=
khaki,linestyle=2):
L_plus_delta:=spacecurve([t,0,14+epsilon],t=0..3,thickness=2,color=
khaki,linestyle=2):
> Vue:=display([Surface,Domaine,Voisinage,Image,L,L_moins_delta,
L_plus_delta,Proj_xz,Verticale_domaine,Horizontale_domaine,Verticale,
Horizontale],orientation=[40,72]):
Vue;

```




Effectuons un zoom vers l'avant pour « mieux voir »

```
> display(Vue,view=[0..3,0..3,0..20],orientation=[45,60]);
```




Visualisons la courbe sur la surface entre les plans d'équations $z = 14 - \epsilon$ et $z = 14 + \epsilon$

```
> Plan_sup:=plot3d([x,y,14+epsilon],x=0..3,y=0..3,style=WIREFRAME,
color=wheat):
Plan_inf:=plot3d([x,y,14-epsilon],x=0..3,y=0..3,style=PATCHNOGRID,
color=wheat):
> display([Voisinage,Domaine,Image,L,L_moins_delta,L_plus_delta,
Proj_xz,Verticale,Verticale_domaine,Horizontale,Horizontale_domaine,
Plan_sup,Plan_inf],Surface,axes=normal,orientation=[45,60]);
```



Effectuons un zoom vers l'avant et *n'affichons pas la surface* afin de mieux visualiser les valeurs d'images qui sont comprises entre ces deux plans.

```
> display([Voisinage,Domaine,Image,L,L_moins_delta,L_plus_delta,
Proj_xz,Verticale,Verticale_domaine,Horizontale,Horizontale_domaine,
Plan_sup,Plan_inf],orientation=[45,80],view=[-2..3,-2..3,0..20]);
```



On voit mieux maintenant que $\forall \varepsilon > 0$, il sera toujours possible d'obtenir un voisinage circulaire de $(1, 2, 0)$ de telle manière que les cotes de toutes les images de ce voisinage seront comprises entre les plans d'équations $z = 14 - \varepsilon$ et $z = 14 + \varepsilon$.

Évaluons finalement cette limite avec Maple.

```
> 'limit'(f(x,y), {x=1,y=2})=limit(f(x,y), {x=1,y=2});
      limit(x^2 + 10, {x=1,y=2}) = 14
```

REMARQUE: Dans le cas d'une fonction où l'image d'un voisinage troué $W_O(a, b)$ ne serait pas plane (le cas où il y aurait des «bosses»), il est clair que la projection dans le plan XZ de l'image de la "frontière de $W_O(a, b)$ " pourrait ne pas contenir le maximum et/ou le minimum de l'image. Dans de tels cas, cette projection dans le plan XZ ne serait pas appropriée pour ce voisinage $W_O(a, b)$. Mais, il n'en demeure pas moins, lorsque la limite existe, qu'il sera toujours possible d'obtenir un $W_O(a, b)$ de telle manière que l'image de ce voisinage soit tout entier compris entre les plans d'équations $z = 14 - \varepsilon$ et $z = 14 + \varepsilon$, et ce, $\forall \varepsilon > 0$.

Cas où la limite n'existe pas

Cas où la limite n'existe pas par défaut

Soit la fonction f définie par $f(x, y) = \frac{x^2 + y^2}{x + y}$.

Puisque le $dom_f = \mathbb{R} \times \mathbb{R} \setminus \{(x, y) \mid y = -x\}$, tout voisinage troué de $(0, 0)$ sera parcouru dans le domaine de f par droite d'équation $y = -x$, la limite n'existe donc pas en $(0, 0)$.


Nous allons montrer qu'obtenir une représentation graphique correcte de la surface définie par $z = f(x, y)$ n'est pas une mince affaire.

On entend par représentation graphique correcte un tracé illustrant clairement les caractéristiques de la surface. Mais justement, nous n'avons actuellement aucune idée de cette surface. Tentons un premier tracé sans trop réfléchir.

```
> f := (x,y) -> (x^2+y^2)/(x+y);
      f := (x,y) ↦  $\frac{y^2 + x^2}{x + y}$ 
```


Obtenons un premier tracé de la surface correspondant à l'équation $f(x) = \frac{x^2 + y^2}{x + y}$ pour $x \in [-10; 10]$ et pour $y \in [-10; 10]$ pour voir.

```
> Surface:=plot3d([x,y,f(x,y)],x=-10..10,y=-10..10,
    axes=boxed,Opts3D,orientation=[-10,80,0]);
```



Obtenons un deuxième tracé de la surface davantage près de $(0, 0)$. Traçons à nouveau avec $x \in [-0.25; 0.25]$ et pour $y \in [-0.25; 0.25]$

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.24,y=-0.25..0.24,
    axes=boxed,Opts3D,orientation=[-10,80,0])
```




(Cliquez sur le graphique et obtenez différentes orientations de cette surface).

Après avoir visualisé différentes orientations de cette surface et compte tenu de la manière dont l'afficheur représente cette surface, il semble que la surface présente une faille le long de la droite $y = -x$ dans le domaine de f . Ce qui n'est pas surprenant.

Avant de faire un troisième tracé, notons les grandes valeurs négatives le long de la faille dans le tracé précédent. Améliorons le tracé de cette surface par un contrôle de l'affichage avec l'option `view`.

```
> display(Surface,view=[-0.25..0.25,-0.25..0.25,-2..2]);
```



Clairement, cette « faille » est une caractéristique du tracé de la surface car le domaine est $\mathbb{R} \times \mathbb{R} \setminus \{(x, y) \mid y = -x\}$ mais l'afficheur a, ici, bien sûr exagéré son apparence et nous observons aussi une certaine courbure de la surface représentative de la fonction f au voisinage de $(0, 0)$.

En fait, le tracé correct de cette surface est une surface dont toutes sections parallèles au plan XY est un cercle quasi-tangent au plan d'équation $y = -x$. En effet, simplifions l'équation $f(x, y) = k$:

```
> Éq_1 := f(x, y) = k;
```

$$\text{Éq}_1 := \frac{x^2 + y^2}{x + y} = k$$

```
> Éq_2 := (x + y) * Éq_1;
```

$$\text{Éq}_2 := x^2 + y^2 = (x + y)k$$

```
> Éq_3 := expand(Éq_2 - (rhs(Éq_2) = rhs(Éq_2)));
```

$$\text{Éq}_3 := -kx - ky + x^2 + y^2 = 0$$

```
> Éq_4 := Student[Precalculus][CompleteSquare](Éq_3, x);
```

$$\text{Éq}_4 := \left(x - \frac{k}{2}\right)^2 - ky + y^2 - \frac{k^2}{4} = 0$$

```
> Éq_5 := Student[Precalculus][CompleteSquare](Éq_4, y);
```

$$\text{Éq}_5 := \left(y - \frac{k}{2}\right)^2 + \left(x - \frac{k}{2}\right)^2 - \frac{k^2}{2} = 0$$

```
> Éq_6 := Éq_5 + (1/2*k^2 = 1/2*k^2);
```

$$\text{Éq}_6 := \left(y - \frac{k}{2}\right)^2 + \left(x - \frac{k}{2}\right)^2 = \frac{k^2}{2}$$

On a donc, pour toute "hauteur" $k \in \mathbb{R} \setminus \{0\}$, nous avons un cercle centré au point $\left(\frac{k}{2}, \frac{k}{2}\right)$ de rayon

$\frac{\sqrt{2} |k|}{2}$ qui exclut les points où $y = -x$. Notons que les centres des cercles s'éloignent


perpendiculairement du plan $y = -x$ lorsque la "hauteur" augmente ($z \rightarrow \pm \infty$).

Mais ce contrôle de l'axe des z est-il judicieusement fait pour qu'on puisse "voir" correctement cette surface ? La surface semble avoir une "courbure" qui a été tronquée par un contrôle insuffisant de la cote. Peaufinons davantage le tracé en diminuant l'intervalle de valeurs de la cote dans l'option `view`.

```
> Surface := plot3d([x, y, f(x, y)], x = -0.25..0.24, y = -0.25..0.24, Opts3D):
```

```
> display(Surface, axes = boxed,
```

```
orientation=[-10,80,0],
view=[-0.25..0.25,-0.25..0.25,-0.18..0.18]);
```



Il y a toujours une représentation de la faille mais, cette fois cela est présenté sous la forme d'un plan $y=-x$ (en fait, un quasi plan) Nous devons contourner en quelque sorte le mécanisme utilisé par la procédure de traçage 3D.

La procédure de traçage a pris en compte des couples (x, y) dans le domaine près de la droite $y=-x$, ce qui donnent de très grandes valeur de z (dans le premier tracé, de l'ordre de $\pm 10^{15}$)


$$y \approx -x \Rightarrow x + y \approx 0 \Rightarrow f(x + y) \approx \frac{2x^2}{\approx 0} \approx \begin{cases} \infty & x + y < 0 \\ -\infty & x + y > 0 \end{cases}$$

et, d'autre part, la procédure de traçage a pris en compte de couples (x, y) dans le voisinage de $(0, 0)$ où la fonction f est partout définie et prenant des valeurs de z près de 0. En conséquence, l'afficheur doit composer en même temps avec des valeurs de cotes très grandes avec des valeurs de cotes près de 0.

Il n'y a pas en 3D l'équivalent en 2D pour empêcher de relier les points donnant des pseudo-asymptotes verticales, c'est-à-dire une option équivalente à l'option 2D `discount=true`. Ce qui permettrait de supprimer de pseudo-plans verticaux.

Dans ce cas, employons l'astuce suivante:


```
> Surface:=plot3d(piecewise(abs(x+y)>0.0025,f(x,y),undefined),x=
-.25..0.24,y=-0.25..0.24,
Opts3D,axes=box,
orientation=[-10,80,0],
view=[-0.25..0.25,-0.25..0.25,-0.18..0.18]);
```




Ce dernier tracé est presque un tracé parfait de la surface. Si le tracé avait été parfait, il aurait été moins évident de visualiser qu'il n'existe aucun voisinage de $(0, 0)$ pour lequel la fonction est toujours définie puisque les points manquants de cette surface le long de l'axe des z (et appartenant au plan $y = -x$) ne peuvent être visibles à l'écran. Un point étant de dimension nulle.

Superposons à ce tracé quelques cercles d'équation $\left(y - \frac{k}{2}\right)^2 + \left(x - \frac{k}{2}\right)^2 = \frac{k^2}{2}$.

```
> a:=k/2:
  b:=k/2:
  r:=sqrt(k^2/2):
  Liste:=[0.001,seq(-k*0.175/20,k=1..20),seq(k*0.175/20,k=1..20)]:
> L:=seq(plots[spacecurve]([a+r*cos(t),b+r*sin(t),k], t = 0 .. 2*Pi,
  color="Burgundy",thickness=2),k=Liste):
> display(Surface,L);
```



```
> display(L,
  orientation=[-10,80,0],
  view=[-0.25..0.25,-0.25..0.25,-0.18..0.18]);
```



Évaluons cette limite et voyons le résultat que l'évaluateur donnera.

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
  Limit( $\frac{x^2 + y^2}{x + y}$ , {x=0,y=0}) = undefined
```

L'évaluateur n'a pas été en mesure de résoudre la simplification demandée. Mais nous, nous savons que cette limite n'existe pas par défaut.

Cas où la limite n'existe pas par approche

Soit la fonction f définie par $f(x, y) = \frac{x^2}{x^2 + y^2}$. Le domaine de la fonction f est $\mathbb{R}^2 \setminus \{(0, 0)\}$. Dans ce cas, la fonction f est toujours définie dans n'importe quel voisinage troué de $(0, 0)$.

Demandons immédiatement à l'évaluateur de calculer la limite de $f(x, y)$ lorsque $(x, y) \rightarrow (0, 0)$.

```
> f:=(x,y)->x^2/(x^2+y^2);
```

$$f := (x, y) \mapsto \frac{x^2}{y^2 + x^2}$$

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

$$\text{Limit}\left(\frac{x^2}{x^2 + y^2}, \{x=0, y=0\}\right) = 0..1$$


L'évaluateur a été en mesure de résoudre la simplification demandée et c'est de cette manière que l'évaluateur nous informe que la limite demandée n'existe pas.

Traçons cette fonction pour $x \in [-0.25, 0.25]$ et pour $y \in [-0.25, 0.25]$.

```
> Opts3D:= grid=[80,80],style=patch,lightmodel=light2,shading=xy:
```

```
> Surface:=plot3d([x,y,f(x,y)],x=-0.25..0.25,y=-0.25..0.25,Opts3D):
```

```
> display(Surface,orientation=[56,68]);
```



Obtenons un meilleur rendu de la surface dans le voisinage de $(0, 0)$ avec un découpage de cette surface en quatre morceaux.

```
> Surface1:=plot3d([x,y,f(x,y)],x=0..0.25,y=0..0.25,Opts3D):
```


```
> Surface2:=plot3d([x,y,f(x,y)],x=-0.25..0,y=0..0.25,Opts3D):
```

```
> Surface3:=plot3d([x,y,f(x,y)],x=-0.25..0,y=-0.25..0,Opts3D):
```

```
> Surface4:=plot3d([x,y,f(x,y)],x=0..0.25,y=-0.25..0,Opts3D):
```

```
> Vue:=display(Surface1|($1..4),transparency=0.25,orientation=[56,68]):
```


```
Vue;
```



Maintenant, on voit bien qu'avec une approche de $(0, 0)$ le long de l'axe des x ($y = 0$), les images $f(x, 0)$ par la fonction f tend vers 1 lorsque $x \rightarrow 0$.

Illustrons cette approche le long de l'axe des x .

```
> Chemin:=spacecurve([t,0,0],t=0..0.25,
    color="DarkOrchid",
    thickness=3):
Image:=spacecurve([t,0,f(t,0)],t=0..0.25,
    grid=[60,60],color="DeepBlue",
    thickness=4):
> display([Vue,Chemin,Image],orientation=[35,65]);
```



Évaluons la limite avec l'approche $y = 0$.


```
> Limit(f(x,0),x=0)=limit(f(x,0),x=0);
    
$$\lim_{x \rightarrow 0} 1 = 1$$

```

Alors, avec cette approche, $f(x, y) = \frac{x^2}{x^2 + y^2}$ se réduit à $\frac{x^2}{x^2}$, soit 1, et donc que la limite de $f(x, y)$ selon ce chemin est 1.


En modifiant l'orientation du graphique précédent, cela nous permet de constater qu'avec une approche le long de l'axe des y , ($x = 0$), les images $f(0, y)$ par la fonction f tend vers 0 lorsque $y \rightarrow 0$.

```
> display(Vue,orientation=[135,65]);
```



Illustrons les images de (x, y) le long de l'axe des y , soit le chemin $x = 0$.

```
> Chemin:=spacecurve([0,t,0],t=0..0.25,
    color="DarkOrchid",
    thickness=3):
Image:=spacecurve([0,t,f(0,t)],t=0..0.25,
    color="DeepBlue",
    thickness=4):
> display([Vue,Chemin,Image],orientation=[135,65]);
```



Évaluons la limite selon le chemin $x = 0$.

```
> Limit(f(0,y),y=0)=limit(f(0,y),y=0);
    lim 0 = 0
    y -> 0
```

Analytiquement, selon le chemin $x = 0$, $f(x, y) = \frac{x^2}{x^2 + y^2}$ se réduit à 0 ($y \neq 0$), et donc que la limite de $f(x, y)$ est 0.

On conclut donc que la limite demandée, la limite de $f(x, y)$ lorsque $(x, y) \rightarrow (0, 0)$, n'existe pas.


Il y a bien sûr d'autres chemins menant à l'origine; par exemple, (x, y) peut atteindre l'origine en suivant la droite $y = 2x$, ou bien en suivant la parabole $y = x^2$, ou bien en suivant la courbe $x = \arcsin(y)$, ou bien en suivant la parabole cubique $y = x^3$, ou bien en suivant le chemin $y = e^x - 1$, etc... Il y a une infinité de chemins passant par $(0, 0)$.

Illustrons les images de (x, y) avec le chemin $y = x$.

```

> Chemin:=spacecurve([t,t,0],t=0..0.25,
    color="DarkOrchid",
    thickness=3):
Image:=spacecurve([t,t,f(t,t)],t=0..0.25,
    color="DeepBlue",
    thickness=4):
> display([Vue,Chemin,Image],shading=zhue,orientation=[15,70]);

```



Évaluons la limite avec l'approche $y = x$.

```

> Limit(f(x,x),x=0)=limit(f(x,x),x=0);

```

$$\lim_{x \rightarrow 0} \frac{1}{2} = \frac{1}{2}$$


Analytiquement, avec cette approche ($y = x$), $f(x, y) = \frac{x^2}{x^2 + y^2}$ se réduit à $\frac{x^2}{x^2 + x^2} = \frac{1}{2}$ et donc, que la limite de $f(x, y)$ selon ce chemin est $\frac{1}{2}$.

Allons-y avec un dernier chemin $y = e^x - 1$.

```

> Chemin:=spacecurve([t,exp(t)-1,0],t=0..0.20,
    color="DarkOrchid",
    thickness=3):
Image:=spacecurve([t,exp(t)-1,f(t,exp(t)-1)],t=0..0.22,
    color="DeepBlue",
    thickness=4):
> display([Vue,Chemin,Image],shading=zhue,orientation=[75,75]);

```



Évaluons la limite avec l'approche $y = e^x - 1$.

> $\text{Limit}(f(x, \exp(x)-1), x=0) = \text{limit}(f(x, \exp(x)-1), x=0);$

$$\lim_{x \rightarrow 0} \frac{x^2}{(e^x - 1)^2 + x^2} = \frac{1}{2}$$

Analytiquement, avec cette approche ($y = e^x - 1$), $f(x, y) = \frac{x^2}{x^2 + y^2}$ se réduit à $\frac{x^2}{x^2 + (e^x - 1)^2}$ et donc,

que la limite de $f(x, y)$ selon ce chemin est $\frac{1}{2}$.